MACAULAY2 - RTG SEMINAR

NOVEMBER 6TH, 2023

Macaulay2 can compute the normalization of a ring or integral closure of an ideal. The goal today is to learn how to use those functions (and we'll also play around with the recently mentioned "prune" function).

Integral closure of rings.

Recall that the normalization / integral closure of a ring R is the set of all elements in the fraction field / total ring of fractions that are integral over R. Try normalizing some rings.

```
R = QQ[x,y]/ideal(x^2-y^3)
S = integralClosure R
```

It turned out that the normalization map is actually stored as a key in `R`. You can access it via:

```
f = R.icMap
```

or

```
f = icMap R
```

Note in this case the normalization of `R` should be a polynomial ring in one variable, `QQ[t]` where `x = t ^ 3, y = t ^ 2`. Our normalization has lots of unnecessary variables. We can get rid of them by the command `prune`

```
T = prune(S)
g = S.minimalPresentationMap
```

The second command give the constructed map `S` $\longrightarrow$ `T`. You can compose the two ring maps by

```
h = g*f
```

Make sure that works.

<u>Exercise #1</u>: *Create a function / method that takes in a (nonnormal) ring* `R` *and outputs the function to the pruned normalization. Try it out on some examples, like* `QQ[x,y,z]/ideal(x*y^2 - z^2)` *or* `QQ[x,y]/ideal(y^2 - x^3 - x^2)` *(a pinch point or a cusp respectively). I called my function* `prunedICMap`*. If you want to try it on something more interesting, check out the first two examples in:*

`https://www.math.utah.edu/~schwede/M2RTG/SeminarExamples11-6-23.m2`

The third example might be fine too, but Macaulay2 is very slow for me when I try to compute the integral closure of that example (it didn't finish after 5 minutes, and then I gave up.)

Conductor ideals.

Given a ring R, the conductor is the largest ideal that is simultaneously an ideal in both R and its normalization $S \supseteq R$ (ie, the same set is an ideal in *both* rings). The ideal c also defines the set of primes Q such that R_Q is non-normal (that is, the *non-normal locus*). It's also defined as

$$c := \mathrm{Ann}_R(S/R).$$

Macaulay2 can compute conductors (of any finite ring map, defined via annihilators as above). In the above you computed some finite ring maps. Let's try computing the conductor of some of those ring maps we worked with above.

Try out the following code in Macaulay2 (we assume the same `R` as above).

```
conductor (icMap R)
```

what's wrong? (it doesn't work in 1.22 at least). Look at the `help conductor` or `code (conductor, RingMap)` (see `if false`) to see if you can find out what's wrong or read below for a hint. Let's try it on a "different" ring where it works.

```
R1 = QQ[x,y,Degrees=>{3,2}]/ideal(x^2 - y^3);
conductor (icMap R1)
```

Ok, let's fix this. We'll make a conductor function that works on all rings with the package `pushForward`. What `pushForward` lets you do is, given a finite ring map `R` → `S`, is view S-modules (like `S^1`) as an `R`-module via restriction of scalars.

For example, you can run

```
M = (pushFwd f)#0
```

to compute `integralClosure R` as a `R`-module. If we want `R` → `S` as a map of `R`-modules, we run:

```
phi = ((pushFwd f)#2)(sub(1, target f))
```

Note `((pushFwd f)#2)` is a *Macaulay2 function* that takes element b of `S` and returns the `S`-module map `R` $\longrightarrow$ `S` sending $1 \mapsto b$. To compute the conductor you would then run.

```
c = ann coker phi
```

Exercise #2: *Write a function (I called mine* `betterConductor`*) which takes a finite ring map, and returns the conductor. See if it works on the examples above. You can also compare it in terms of timing, to this.*

```
E = QQ[x,y,z]/ideal(x^11-z^11-y^2*z^9);
time conductor icMap E
time betterConductor icMap E
```

I would recommend you `restart` *when doing timing (as Macaulay2 cache's lots of things). To make it faster, instead of using* `pushFwd`*, use the commented out strategy from* `(conductor, RingMaps)`*.*

Integral closure of ideals.

Given an ideal J in a ring R, we can compute its integral closure too. Algebraically, this is the set of solutions to equations:

$$X^n + f_1 X^{n-1} + \cdots + f_{n-1} X + f_n = 0$$

where $f_i \in J^i$. It can alternately be viewed (under moderate hypotheses) as the the largest ideal with the same normalized blowup as J (at least if one tracks the induced relatively anti-ample exceptional divisor too, corresponding to $\mathcal{O}(-1)$). Some fun examples to try this on.

```
R = QQ[x,y];
J = ideal(x^9, y^9);
integralClosure J
```

or

```
R = QQ[x,y,z];
I = ideal(x^4+y^4+z^4+x*y);
J = ideal jacobian I
integralClosure J
```